
Introduction

The objective of this paper is to provide an introduction to dynamic programming that emphasizes intuition over mathematical rigor.1 The discussion is built around a set of examples that cover
a number of key problems from economics. In contrast to other brief introductions to dynamic programming, such as King (2002), the paper integrates the use of a computer algebra system.2 Maple
program code is provided for every example.3 Based on classroom experience, this allows readers
not only to easily replicate and extend all problems but also to see more clearly their structure. The
problems in this paper are fully accessible to undergraduates that are familiar with the Lagrangian
multiplier method to solve contrained optimization problems.
Dynamic programming has strong similarities with optimal control, a competing approach to
dynamic optimization. Dynamic programming has its roots in the work of Bellman (1957), while
optimal control techniques rest on the work of the Russion mathematician Pontryagin and his
coworkers in the late 1950s.4 While both dynamic programming and optimal control can be applied to discrete time and continuous time problems, most current applications in economics appear
to favor dynamic programming for discrete time problems and optimal control for continous time
problems. To keep the discussion reasonably simple, this paper will only deal with discrete time
dynamic programming problems. These have become very popular in recent years in macroeconomics. However, the discussion will not be limited to macroeconomics but will try to convey the
idea that dynamic programming has applications in other settings as well.
The paper is organized as follows. The next section provides some motivation for the use of
dynamic programming techniques. This is followed by a discussion of finite horizon problems. Both
numerical and non-numerical problems are treated. The subsequent section covers infinite horizon
problems, again using both numerical and non-numerical examples. The last section before the
concluding comments solves a simple stochastic infinite horizon problem of the real business cycle
variety.

A Motivating Example

The basic principle of dynamic programming is best illustrated with an example. Consider
for that purpose the problem of an oil company that wants to maximize profits from an oil well.
1

Readers interested in more formal mathematical treatments are urged to read Sargent (1987), Stokey and Lucas
(1989) or Ljungqvist and Sargent (2000).
2
The book by Miranda and Fackler (2002) emphasizes numerical techniques to solve dynamic programming problems. The software package Matlab is used extensively. Adda and Cooper (2003) put less emphasis on computatioal
aspects, but nonetheless make available a fair amount of computer code in both Matlab and Gauss.
3
The brief overview of discrete time dynamic programming in Parlar (2000) was very helpful to the author because
of its focus on coding in Maple.
4
Kamien and Schwartz (1981) contains one of the most thorough treatments of optimal control. Most major
textbooks in mathematical methods for economists also contain chapters on optimal control.

Revenue at time t is given as
Rt = pt ut ,
where p is the price of oil and where u is the amount of oil that is extracted and sold. In dynamic
programming applications, u is typically called a control variable. The company’s cost function is
quadratic in the amount of oil that is extracted,
Ct = 0.05u2t .
The amount of oil remaining in the well follows the recursion or transition equation
xt+1 = xt − ut ,
where x is known as a state variable in dynamic programming language. Since oil is a non-renewable
resource, pumping out oil in the amount of u at time t means that there is exactly that much less
left in the oil well at time t + 1. We assume that the company applies the discount factor β = 0.9
for profit streams that occur in the future. We also assume that the company intends to have the
oil well depleted in 4 years, which means that x4 = 0. This is known as a boundary condition in
dynamic programming. Given these assumptions, the central question to be solved is how much oil
should be pumped at time t, t = 0, ..., 3 to maxime the discounted profit stream.
If one had never heard of dynamic programming or optimal control, the natural way to solve
this problem is to set up a Lagrangian multiplier problem along the following lines,
max

L(x, u, λ) =

3
X
t=0

β t (Rt − Ct ) +

4
X
t=1

λt (xt − xt−1 + ut−1 ),

where x = (x1 , x2 , x3 ), with x0 given exogenously and x4 = 0 by design, u = (u0 , u1 , u2 , u3 ), λ =
(λ1 , λ2 , λ3 , λ4 , ) and where the constraints simply specify that the amount of oil left at period t
is equal to the amount in the previous period minus the amount pumped in the previous period.
Using standard techniques the above Lagrangian problem can be solved for the decision variables
x and u and the Lagrangian multipliers λ in terms of the exogenous variables {p0 , p1 , p2 , p3 , x0 }.
The Maple commands to solve this problem are given as follows:
restart: with(linalg): Digits:=(3):
f:=sum(0.9^t*(p[t]*u[t]-0.05*u[t]^2),t=0..3);
x[0]:=x0: x[4]:=0:
g:=sum(lambda[t]*(x[t]-x[t-1]+u[t-1]),t=1..4);
L:=f+g;
L_grad:=grad(L,[seq(x[t],t=1..3),seq(u[t],t=0..3),
seq(lambda[t],t=1..4)]);

solve({seq(L_grad[i],i=1..11)},{seq(x[t],t=1..3),
seq(u[t],t=0..3),seq(lambda[t],t=1..4)});
assign(%):
For example, the optimal amounts of oil to pump in periods zero to four are given as
u0 = .212x0 + 7.88p0 − 2.12p1 − 2.12p2 − 2.12p3
u1 = .235x0 − 2.35p0 + 7.65p1 − 2.35p2 − 2.36p3
u2 = .262x0 − 2.62p0 − 2.62p1 + 7.38p2 − 2.62p3
u3 = .291x0 − 2.91p0 − 2.91p1 − 2.91p2 + 7.10p3 .
If one now specifies a particular sequence of prices and the value of x0 , then a complete numerical
solution results. For example, assuming {x0 = 1000, p0 = 20, p1 = 22, p2 = 30, p3 = 25} the
additional Maple commands
x0:=1000: p[0]:=20: p[1]:=22: p[2]:=30:
evalf([seq(x[t],t=0..4),seq(u[t],t=0..3),
seq(lambda[t],t=1..4)]);

p[3]:=25:

generate the following numerical solution for x, x 4 , u, λ,
[1000,794,566,260,0,206,227,308,260,0.60,0.60,0.60,0.60].
Given the relative simplicity with which the above problem can be solved with standard techniques, one may wonder where there is a room for a diﬀerent approach. The clue to this puzzle lies
in the simple fact that one cannot at time 0 pretend to know all future prices or actual amounts of
oil that are pumped. For example, it may be that oil rigs break down or workers are on strike such
that the optimal amount of oil simply cannot be pumped in a given year. Yet it should be clear
that a change in any of the future p or x means that the above sequence of x and u is not optimal
any longer. The whole problem needs to be recalculated subject to the new p and the known values
of x. Rather than recalculating the problem whenever something unexpected happens to either p
or x, one could try to find a decision rule or policy function for the control variable u that can be
followed regardless of random shocks to price or state variable x. That is the basic idea of dynamic
programming.

3
3.1

Finite Horizon Problems
The Motivating Example Again

We will now recalculate the above problem using dynamic programming techniques. The objective is, as before, to maximize
3
X
t=0

β t (pt ut − 0.05u2t )

subject to the transition equations
xt+1 = xt − ut

t = 0, 1, 2, 3

and the boundary condition
x4 = 0.
We assume again that the company applies the discount factor β = 0.9.
The problem is solved with the help of the Bellman equations
Vt = max(pt ut − 0.05u2t ) + 0.9Vt+1 (xt − ut ),
ut

t = 0, 1, 2,

(1)

where Vt is known as the value function in dynamic programming terminology. We note that the
value function V appears both on the left and on the right of the Bellman equation, although with
a diﬀerent time subscript. The solution process works backward. We start with the boundary
condition and the value function for t = 3 and then solve the problem recursively in the backward
direction toward period 0.
The starting point is the value function for t = 3. Since x4 = 0, the transition equation for
period 3 is given as
x4 = 0 = x3 − u3 .
The transition equation for t = 3 conveniently provides us with the optimal decision rule for the
control variable u in period 3,
u3 = x3 .

(2)

Equation 2 simply says to extract and sell all the oil that remains at the beginning of t = 3 in
that same period. Compared to the rule that we obtained from applying the Lagrange multiplier
method in the previous section, equation 2 is much simpler because it is independent of any price
variables or any other variables dated t < 3. It is an application of the principle of optimality that

underlies dynamic programming: we can do no better than just follow the derived optimal policy
or decision rule regardless of what happens to control and state variables in earlier periods.
To obtain the optimal decision rules for the control variable in the periods prior to t = 3, we
will make use of the recursive nature of Bellman equation. One period at a time, we will move
backward in time until we have found the decision rule for t = 0.
The first step in finding the decision or policy rule for setting the control variable in t = 2
consists of substituting the optimal policy function for t = 3 into the value function for t = 3,5
V3 = (p3 u3 − 0.05u23 ) + 0.9V4 (x3 − u3 ).
By equation 2, this simplifies to
V3 = p3 x3 − 0.05x23 .

(3)

Next, we substitute equation 3 into the Bellman equation for period 2
V2 = max (p2 u2 − 0.05u22 ) + 0.9V3
u2

to obtain
V2 = max(p2 u2 − 0.05u22 ) + 0.9(p3 x3 − 0.05x23 ).
u2

(4)

Before we can maximize the right-hand side of equation 4 with respect to u2 we need to consider
that x3 is connected with u2 via the transition equation
x3 = x2 − u2 .

(5)

V2 = max(p2 u2 − 0.05u22 ) + 0.9[p3 (x2 − u2 ) − 0.05(x2 − u2 )2 ].

(6)

Substituting equation 5 into 4 we obtain

u2

Now, only the decision variable u2 and variables that are known or exogenous at time t = 2 are in
equation 6. The right-hand side of equation 6 can be maximized with respect to the control variable
u2 ,
¤
£
d (p2 u2 − 0.05u22 ) + 0.9(p3 (x2 − u2 ) − 0.05(x2 − u2 )2 )
=0
du2
p2 − 0.19u2 − .9p3 + 0.09x2 = 0.
5
The value function is equal to the Bellman equation in (1), except that all occurrences of u3 are replaced by the
optimal value of u3 , which is x3 according to equation 2.

Solving the above first-order condition for u2 gives the optimal policy or decision rule for period 2,
u2 = 0.4737x2 + 5.263p2 − 4.737p3 .

(7)

Similar to the decision rule in equation 2, this policy rule for the control variable is simpler than
the correspondig rule from the Lagrangian multiplier approach in the last section. There is also no
need to revise this rule in the light of unexpected shocks to price or the state variable x prior to
t = 2. This completes the calculations for t = 2.
To find the optimal policy rule for period 1,we need to insert equation 7 into the value function
for period 2 to get an expression similar to equation 3. After some simplifications, the value function
for period 2 can be written as
V2 = .474(p2 + p3 )x2 + 2.636p22 − 4.734p2 p3 − 0.02368x22 + 2.132p23 .
The value function for period 2 needs to be substituted into the Bellman equation for period 1
V1 = max (p1 u1 − 0.05u21 ) + 0.9V2 .
u1

One obtains
V1 = max(p1 u1 − 0.05u21 ) + 0.9(.474(p2 + p3 )x2
u1

+2.636p22 − 4.734p2 p3 − 0.02368x22 + 2.132p23 )

(8)

Making use of the transition equation
x2 = x1 − u1

(9)

in equation 8, one can maximize its right-hand side with respect to u1 ,
d[(p1 u1 − 0.05u21 ) + 0.9(.474(p2 + p3 )(x1 − u1 )
+2.636p22 − 4.734p2 p3 − 0.02368(x1 − u1 )2 + 2.132p23 )]
du1

= 0.

Solving the resulting first-order condition for u1 one obtains
u1 = .2989x1 + 7.013p1 − 2.989p2 − 2.989p3 .
This is the optimal policy or decision rule for period 1. We note that it has some similarities with
the equation that determines the optimal value of u1 from the Lagrangian multiplier approach.
Yet, it does have fewer terms.
The last step in the solution of the optimal control problem is to obtain the decision or policy
rule for period 0. The procedure is the same as before. Insert the optimal decision rule for period

1 into the Bellman equation for period 1 and simplify to obtain
V1 = −.01494x21 + 3.006p22 − 2.989p2 p3 + .2989(p1 + p2 + p3 )x1
+2.556p23 + 3.506p21 − 2.989p1 (p2 + p3 )

Next, the above value function is inserted into the Bellman equation for period 0, which is given as
V0 = max (p0 u0 − 0.05u20 ) + 0.9V1 .
u0

(10)

After substituting out all occurrences of x1 in the above equation with (x0 − u0 ), the right-hand
side of equation 10 is maximized with respect to u0 . One obtains
u0 = .2120x + 7.880p0 − 2.119p1 − 2.120p2 − 2.120p3 .
This decision rule for period 0 is the same as the one derived in the previous section using Lagrangian
multiplier techniques.
The above results can be calculated by Maple with the help of the following code:
restart: N:=3: Digits:=4:
V[N]:=unapply(p[N]*x-0.05*x^2,x);
for t from N-1 by -1 to 0 do
c[t]:=p[t]*u-0.05*u^2+0.9*V[t+1](x-u);
deriv[t]:=diff(c[t],u);
mu[t]:=evalf(solve(deriv[t],u));
V[t]:=unapply(evalf(simplify(subs(u=mu[t],c[t]))),x);
od;
It should be apparent that the recursive structure of the dynamic programming problem makes
it easy to extend the optimization to a larger number of periods.

3.2

A Non-Numerical Example

Much of the work in economics is about solving optimization problems without specific numerical
entries. The next example applies the dynamic programming framework in this environment.
Consider the maximization of utility of a consumer that does not work but lives oﬀ a wealth
endowment (a). Assume that utility depends on the discounted logged value of consumption (c)
from period zero to two,
max U =

2
X
t=0

7

β t ln ct

where the discount factor is
β=

1
1+r

and where r is the relevant interest rate. Consumption is the model’s control variable and the
transition equations are given by
at+1 = (1 + r)at − ct

t = 0, 1, 2.

The model’s boundary condition is
a3 = 0,
which says that wealth is supposed to be depleted at the end of period 2. There are no bequests.
The task is to find the optimal consumption sequence.
We start the solution process in the final period and work our way backward. Applying the
boundary condition to the transition equation for period t = 2 results in
0 = (1 + r)a2 − c2 .
This can be solved for c2 , the control variable for t = 2, to get
c2 = (1 + r)a2 .

(11)

This is the decision rule for period t = 2. We still need the decision rule for periods t = 1 and t = 0.
To find these, we apply the Bellman equation
Vt = max ln ct + βVt+1 .
ct

The Bellman equation for period 1 requires the value function of period 2. The latter is equal to
the above Bellman equation for t = 2, with c2 replaced by its maximized value from equation 11
and with V3 = 0 because wealth is exhausted at the end of t = 2,
V2 = ln c2 = ln [(1 + r)a2 ] .
This value function can now be inserted into the Bellman equation for period 1,
V1 = max ln c1 + βV2
c1

to obtain
V1 = max ln c1 + β ln [(1 + r)a2 ] .
c1

Before we can maximize the right-hand side of the Bellman equation with respect to c1 , a2 has to
be replaced using the transition equation for t = 1,
a2 = (1 + r)a1 − c1 .
The Bellman equation for period 1 is now of the form,
V1 = max ln c1 + β ln [(1 + r) [(1 + r)a1 − c1 ]] .
c1

and its right-hand side can be maximized with respect to the control variable c1 . The first-order
condition and c1 are given as
d [ln c1 + β ln (1 + r) + β ln [a1 (1 + r) − c1 ]]
−a1 (1 + r) + c1 (1 + β)
= 0,
=
dc1
c1 [−a1 (1 + r) + c1 ]

c1 = (1 + r)

a1
.
1+β

(12)

The decision rule for t = 0 still needs calculating. It requires the value function for t = 1. This value
function is equal to the Bellman equation for period 1, with c1 replaced by the optimal decision
rule, equation 12,
V1 = ln

a1 (1 + r)2 β
a1 (1 + r)
+ β ln
.
1+β
1+β

Hence, the Bellman equation for t = 0 is given as
V0 = max ln c0 + βV1
c0

V0 = max ln c0 + β ln
c0

a1 (1 + r)
a1 (1 + r)2 β
+ β 2 ln
1+β
1+β

Substituting out a1 with
a1 = (1 + r)a0 − c0
gives
¶
¶
µ
µ
(1 + r)a0 − c0
2 ((1 + r)a0 − c0 ) β
2
+ β ln (1 + r)
V0 = max ln c0 + β ln (1 + r)
c0
1+β
1+β
Maximizing the right-hand side with respect to the control variable c0 ,results in the following

first-order condition and optimal value for c0
dV0
−a0 (1 + r) + c0 (1 + β + β 2 )
=0
=
dc0
c0 (−a0 (1 + r) + c0 )

c0 = (1 + r)

a0
.
1 + β + β2

(13)

A comparison of the consumption decision rules for t = 0, 1, 2 reveals a pattern,

or, more generally, for t = 0, ..., n

at (1 + r)
ct = P2−t i ,
i=0 β

t = 0, 1, 2,

at (1 + r)
at
at
ct = Pn−t i = Pn−t i = Pn−t+1 i ,
β i=0 β
β
i=0 β
i=1

t = 0, ..., n.

(14)

The ability to recognize and concisely describe such a pattern plays a key role in applications of
dynamic programming in economics, in particular those applications that deal with infinite horizon
problems. They will be discussed next.
The Maple program that solves the above consumption problem follows.
restart: N:=2: Digits:=4:
V[N]:=unapply(log((1+r)*a),a);
# V[N] is made a function of a
# this value function follows from the boundary condition
for t from N-1 by -1 to 0 do
VV[t]:=log(c)+beta*V[t+1]((1+r)*a-c);
# Since V is a function of a, the term in parenthesis after
# V[t+1] is substituted for a everywhere a appears in V[t+1]
deriv[t]:=diff(VV[t],c);
mu[t]:=solve(deriv[t],c);
V[t]:=unapply(simplify(subs(c=mu[t],VV[t])),a);
od;

4

Infinite Horizon Problems

A question often posed in economics is to find the optimal decision rule if the planning horizon
does not consist of a fixed number of n periods but of an infinite number of periods. Examing the
limiting case n → ∞ may also be of interest if n is not strictly infinite but merely large because
taking the limit leads to a decision rule that is both simplified and the same for every period. Most

often that is preferred when compared to a situation where one has to deal with a distinct and
rather complex rule for each period from t = 0 to t = n. We shall illustrate this point with the
consumption problem of the last section.

4.1

A Problem Without Numerics

We want to modify the optimal consumption decision rule given in equation 14 so it applies for
an infinite time horizon. For taking the limit, the first part of equation 14 is most useful,
at (1 + r)
lim ct = lim Pn−t i .
n→∞
i=0 β

n→∞

In taking the limit, one needs to make use of the fact that
∞
X

βi =

i=0

1
.
1−β

Application of this rule yields
lim ct = at (1 + r) (1 − β) = at r.

n→∞

The decision rule for consumption in period t simplifies to the intuitively obvious: consume in t
only what you get in interest from your endowment in t. The associated transition function for
wealth is given as
at+1 = (1 + r)at − ct = (1 + r)at − at r = at ,
that is, wealth will remain constant over time.
This example has illustrated one way of arriving at optimal decision rules that apply to an
infinite planning horizon: simply solve the problem for a finite number of periods, identify and
write down the pattern for the evolution of the control variable, and take the limit n → ∞. It
should be obvious that the same procedure can be applied to identify the value function for time
period t. To illustrate this point, recall the sequence of value functions for t = 1, 2
V2 = ln [(1 + r)a2 ]

V1 = ln

a1 (1 + r)2 β
a1 (1 + r)
+ β ln
.
1+β
1+β

(15)

(16)

It is easy although tedious to verify that the value function for t = 0 is given as
V0 = ln

a0 (1 + r)
a0 (1 + r)2 β
a0 (1 + r)3 β 2
2
+
β
ln
+
β
ln
.
1 + β + β2
1 + β + β2
1 + β + β2

(17)

One can proceed now as in the case of the consumption rule. The following pattern emerges from
equations 15 to 17,
Vt =

n−t
X

β i ln

i=0

at (1 + r)i+1 β i
.
Pn−t i
i=0 β

(18)

It is possible to check the correctness of this pattern by letting a program like Maple calculate the
sequence of the Vt . For equations 15 through 17, the Maple code is given as
n:=2:
seq(V[t]=sum(beta^i*log(a[t]*(1+r)^(i+1)*beta^i/
(sum(beta^(i),i=0..n-t))),i=0..n-t),t=0..n);
After one has verified that the formula is correct, one can take the limit n → ∞
lim Vt = lim

n→∞

n→∞

n−t
X

β i ln

i=0

at (1 + r)i+1 β i
.
Pn−t i
i=0 β

Since interest centers on finding an expression involving the variable at , the idea is to isolate at .
This can be done as follows
Ã
!
n−t
X
(1 + r)i+1 β i
i
β ln at + ln Pn−t i
lim Vt = lim
n→∞
n→∞
i=0 β
i=0
or

lim Vt = lim

n→∞

n→∞

n−t
X

β i ln at + lim

n→∞

i=0

The last expression can be rewitten as
lim Vt = lim

n→∞

n→∞

n−t
X

n−t
X
i=0

β i ln

(1 + r)i+1 β i
Pn−t i .
i=0 β

β i ln at + η

(19)

i=0

where η is an expression independent of a. Taking the limit leads to
lim Vt =

n→∞

1
1
ln at + η = (1 + ) ln at + η.
1−β
r

(20)

With the help of a computer algebra program one could circumvent equation 18 and the following

algebra altogether by going from equation 17 immediately to an expression like 19 for n = 2. The
sequence of Maple commands below takes equation 17, expands it, and isolates the terms in a.
assume(beta>0,r>0);
assume(a[0],real);
V[0]:=expand(V[0]);
collect(V[0],log(a[0]));
The output of these commands is
V0 = (1 + β + β 2 ) ln a0 + ζ

(21)

where ζ consists of a number of terms in β and r. One could do the same with equation 16 if one
had trouble moving from equation 21 immediately to equations 19 and then 20.

4.2

A Numerically Specified Problem

Infinite horizon problems that are numerically specified can be solved by iterating on the value
function. This will be demonstrated for the above consumption optimization problem. The task is
to find the optimal decision rule for consumption regardless of time t. For that purpose we employ
the Bellman equation without time subscripts,
V (a) = max ln c + βV (a0 ),
c

where a and a0 denote current and future value of the state variable, respectively. The time
subscripts are eliminated in the Bellman equation to reflect the idea that the value function V (a)
needs to satisfy the Bellman equation for any a. The iteration method starts with an initial guess
of the solution V (a), which we will identiy as V (0) (a). Guessing a possible solution is not as diﬃcult
as it sounds. Each class of problem typically has associated with it a general form of the solution.
For example, in the current case we know from solving the consumption problem for t = 2, that
the value function will somehow depend on the log of a
V (0) (a) = A + B ln a.
We will assume that β = 0.95.
4.2.1

A numerically specified initial guess

We convert the initial guess of the value function into numerical form by assuming A = B = 1,
V (0) (a) = ln a.

The initial guess V (0) (a) is substituted on the right-hand side of the Bellman equation
V (1) (a) = max ln c + 0.95 ln a0 .
c

Next, we make use of the tranition equation
a0 = (1 + 0.05263)a − c
to substitute out a0
V (1) (a) = max ln c + 0.95 ln (1.05263a − c) .
c

We maximize the right-hand side of V (1) (a) with respect to the control variable c,
1
0.95
dV (1) (a)
= −
= 0.
dc
c (1.05263a − c)
which gives c = 0.53981a. Upon substituting the solution for c back into V (1) (a) one obtains
V (1) (a) = ln(0.53981a) + 0.95 ln (1.05263a − (0.53981a))
or after simplifying
V (1) (a) = −1.251 + 1.95 ln a.
Since
V (1) (a) 6= V (0) (a),
we need to continue to iterate on the value function. For that purpose, we substitute the new guess,
V (1) (a), into the Bellman equation and get
V (2) (a) = max ln c + 0.95(−1.251 + 1.95 ln a0 ).
c

Substitution of the transition equation for a0 gives
V (2) (a) = max ln c + 0.95(−1.251 + 1.95 ln(1.05263a − c))
c

which simplifies to
V (2) (a) = max ln c − 1.1884 + 1.8525 ln (1.05263a − c) .
c

Maximizing the right-hand side of V (2) (a) with respect to c one obtains c = 0.36902a. Substituting
the optimal policy rule for c back into V (2) (a) gives
V (2) (a) = ln 0.36902a − 1.1884 + 1.8525 ln (1.05263a − 0.36902a)
which simplifies to
V (2) (a) = −2.8899 + 2.8525 ln a.
Since
V (2) (a) 6= V (1) (a),
another iteration is needed. Instead of continuing the above process of interating on the value
function by hand, we make use of a Maple program to iterate until convergence is achieved,
restart: Digits:=8; beta:=0.95:
V[0]:=a->ln(a);
for n from 0 to 200 do
VV[n]:=(a,c)->ln(c)+beta*V[n]((1+((1/beta)-1))*a-c):
diff(VV[n](a,c),c):
c_opt:=evalf(solve(%,c));
subs(c=c_opt,VV[n](a,c)):
V[n+1]:=unapply(evalf(simplify(%)),a);
od;
After about 200 iterations with eight digit accuracy, the value function reaches
V (a) = 19.9994 ln a − 58.886
and the corresponding policy function for the control variable
c = 0.05263a.
These results are eﬀectively identical to the ones that can be obtained from equation 20 and the
corresponding decision rule for c.6 Specifically, equation 20 predicts the value function to be
V (a) = 20 ln a + η
6
If the calculations are done with fewer digits accuracy, there is more of a discrepancy in results regardless of the
number of iterations.

whereas the corresponding decision rule for c would be
c = 0.05263a.
4.2.2

A numerically unspecified initial guess

We start with the numerically unspecified guess
V (0) (a) = A + B ln a.
where A and B are constants to be determined. In this case, the value function will not be iterated
on. This is of significant advantage if one cannot rely on the help of a computer algebra system.
The solution process starts with the equation
V (a) = max ln c + 0.95V (0) (a0 )
c

and the substitution of the initial guess
V (a) = max ln c + 0.95(A + B ln a0 ).
c

Next, the transition equation is used to substitute out the state variable a0
V (a) = max ln c + 0.95(A + B ln(1.05263a − c)).
c

The right-hand side of V (a) is maximized with respect to the control variable c,
d (ln c + 0.95(A + B ln(1.05263a − c)))
=0
dc
which results in
c=

21.0526a
.
20 + 19B

The optimal c is substituted back into V (a), which, upon expansion, results in
V (a) = 3.047 + ln a − ln (20 + 19B) + .95A + 2.846B + .95B ln a
+.95B ln B − .95B ln (20 + 19B) .
The steps up to this point have been standard. The solution process from here on is diﬀerent.
If V (0) (a) had only one unknown parameter, say A, we could substitute V (0) (a) for V (a) on the
left side of the equation and then solve for the unknown value of A. With two unknown parameters,
A and B, to solve for, this method is not useful. An alternative solution procedure rests on the
comparison of coeﬃcients. In particular, one can compare the coeﬃcient of ln a in V (a) to B,

the coeﬃcient of ln a in V (0) (a), and, later, once B is known, one can compare the value of the
remainder of V (a) to the parameter A in V (0) (a).
To proceed with the coeﬃcient comparison, we collect the terms in ln a
V (a) = 3.047 + (1 + 0.95B) ln a − ln (20 + 19B) + .95A + 2.846B
+.95B ln B − .95B ln (20 + 19B) .
and set the coeﬃcients of ln a in V (a) and V (0) (a) equal to each other,
(1 + 0.95B) = B.
This results in B = 20. To solve for the parameter A, we substitute B = 20 into both V (a) and
V (0) (a), set the two equal, and solve for A,
−2.9444 + 20.0 ln a + .95A = A + 20 ln a
The solution is A = −58.888. With the parameters A and B known, the policy function for c is
given as
c = 0.05263a
and the value function is
V (a) = −58.888 + 20.0 ln a.
Both match the ones obtained in earlier sections with other solution methods.
The Maple commands to replicate the above solution method are
restart: Digits:=8:
assume(a>0):
beta:=0.95:
V[0]:=(a)->A+B*log(a);
VV:=(a,c)->ln(c)+beta*V[0]((1+((1/beta)-1))*a-c);
diff(VV(a,c),c);
c_opt:=evalf(solve(%,c));
subs(c=c_opt,VV(a,c));
V[1]:=collect(expand(simplify(%)),ln(a));
B:=solve(diff(V[1],a)*a=B,B);
V0:=A+B*log(a);
A:=solve(V[1]=V0,A);
c_opt;V0;
17

5

A Stochastic Infinite Horizon Example

Dynamic programming problems can be made stochastic by letting one or more state variables
be influenced by a stochastic disturbance term. Most of the applications of dynamic programming
in macroeconomics follow this path. One example is provided in this section.
We will consider the basic real business cycle (RBC) model discussed in King (2002), which
is very much related to Kydland and Prescott (1982) and Long and Plosser (1983). Many RBC
models are variations of this basic model.
The problem is to maximize the expected discounted sum of present and future utility, where
utility is assumed to be a logarithmic function of consumption (C) and leisure (1 − n)
max E0

∞
X
y=0

β t [ln Ct + δ ln(1 − nt )] ,

where n represents labor input and available time is set at unity. Maximization is subject to
the market clearing constraint that output (y) equals the sum of the two demand components,
consumption and investment
yt = Ct + kt ,
where investment equals the capital stock (k) because capital is assumed to depreciate fully in one
period. There are two transition equations, one for each of the two state variables, output and the
state of technology (A),
yt+1 = At+1 ktα n1−α
t
At+1 = Aρt eεt+1 .
The transition equation of y is a dynamic Cobb-Douglas production function and the transition
equation of A follows an autoregressive process subject to a disturbance term (ε). C, n, and k
are the problem’s control variables. To make the problem easier to follow, we use numbers for the
parameters, β = 0.9, α = 0.3, δ = 1, ρ = 0.8.
The Bellman equation for this problem is given as
V (y, A) = max ln Ct + ln(1 − n) + βEV (0) (y, A),
C,k,n

where E is the expectations operator. The Bellman equation can be simplified by substituting for
C and, hence, reducing the problem to two control variables
V (y, A) = max ln(y − k) + ln(1 − n) + 0.9EV (0) (y, A).
k,n

The next step in the solution process is to find an initial value function. In the earlier consumption

problem, which was similar in terms of the objective function but had only one state variable (a),
we found that the value function was of the type
V (a) = A + B ln a.
Since there are two state variables in the present problem, we try the analogous solution
V (0) (y, A) = Z + G ln y + H ln A.
From here on, the problem follows the methodology used for the previous section. Substituting
V (0) (y, A) into the Bellman equation gives
V (y, A) = max ln(y − k) + ln(1 − n) + 0.9E (Z + G ln y + H ln A) .
k,n

Before the right-hand side of V (y, A) can be maximized, y and A that appear in the expectations
term are substituted out by their transition equations,
¡
¡
¢
¡
¢¢
V (y, A) = max ln(y − k) + ln(1 − n) + 0.9E Z + G ln A0.8 eε k 0.3 n0.7 + H ln A0.8 eε .
k,n

The right-hand side is expanded and then maximized with respect to both k and n. This results in
two first-order conditions that need to be solved simultaneously for the two control variables. The
solution of the maximization process is given as
k=

0.27Gy
,
1 + 0.27G

n=

0.63G
.
1 + 0.63G

These solutions are substituted back into the Bellman equation and the expectations operator is
employed to remove the disturbance terms,
µ
¶
µ
¶
0.27Gy
0.63G
V (y, A) = ln y −
+ ln 1 −
1 + 0.27G
1 + 0.63G
µ
µ
¶
¶
0.27Gy
0.63G
+0.9 Z + G 0.8 ln A + 0.3 ln
+ 0.7 ln
+ H0.8 ln A .
1 + 0.27G
1 + 0.63G
As in the consumption example, coeﬃcients are compared between V (y, A) and V (0) (y, A) to solve
for the unknown values of Z, G, and H. To make a coeﬃcient comparison feasible, we expand the
right-hand side of V (y, A), then simplify, and finally, collect the ln y and ln A terms to get
V (y, A) = 0.72(G + H) ln A + (1 + 0.27G) ln y + θ

where
θ = 9.21 − .9G ln G − .27G ln (100 + 27G) − ln (100 + 27G)
− ln (100 + 63G) + .9Z − .63G ln (100 + 63G) + 3.5G.
Setting the coeﬃcients of ln y in V (y, A) and V (0) (y, A) equal to each other, we obtain the determining equation
(1 + 0.27G) = G,
which gives G = 1.37. Similarly comparing the coeﬃcients of ln A between V (y, A) and V (0) (y, A),
0.72(G + H) = H,
gives H = 3.52. Making use of the solutions for G and H in V (y, A), the value of Z can be
determined from the equation
V (y, A) − 1.37 ln y − 3.52 ln A = Z
as Z = −20.853.
With the values of Z, G, and H known, the value function that solves the problem is given as
V (y, A) = −20.853 + 1.37 ln y + 3.52 ln A,
while the policy functions or decision rules for k, n, and C are
k = 0.27y,

n = 0.463

C = 0.73y.

We note that the addition of a stochastic term to the variable A has no eﬀect on the optimal
decision rules for k, n, and C. This result is not an accident but, as discussed at the beginning of
this paper, the key reason why dynamic programming is preferred over a traditional Lagrangian
multiplier method in dynamic optimization problems: unpredictable shocks to the state variables
do not change the optimal decision rules for the control variables.7
The above solution process can be automated and processed with other parameters if one uses
the Maple commands below,
7

The stochastic nature of state variables do aﬀect how the state variables move through time. Diﬀerent sequences
of shocks to the state variables produce diﬀerent time paths even with the same decision rules for the control variables.
From a large number of such stochastic time paths, RBC researchers calculate standard deviations or other moments
of the state and control variables and compare them to the moments of actually observed variables to check their
models. Compare Adda and Cooper (2003) on this.

restart:
C:=y-k;
yy:=AA*k^alpha*n^(1-alpha);
AA:=A^rho*exp(epsilon);
V0:=ln(C)+delta*ln(1-n)+beta*(Z+G*ln(yy)+H*ln(AA));
delta:=1.; beta:=0.9; rho:=0.8; alpha:=0.3;
assume(y>0,A>0,k>0,n>0):
f1:=expand(V0);
k_opt:=((diff(f1,k)))=0;
n_opt:=diff(f1,n)=0;
solve({k_opt,n_opt},{n,k});
assign(%);
epsilon:=0: expand(f1); f2:=simplify(%);
f3:=collect(%,ln(y));
f4:=collect(%,ln(A));
G:=solve(diff(f4,y)*y=G,G);
H:=solve(diff(f4,A)*A=H,H);
ZZ:=solve(f4-G*ln(y)-H*ln(A)=Z);
V:=ZZ+G*ln(y)+H*ln(A); k; n;

6

Summary and Conclusion

The purpose of this paper has been to introduce dynamic programming techniques by way of
example with the help of the computer algebra system Maple. The emphasis of this introduction
is to build confidence and intuition. The examples are solved one step at a time, with suﬃcient
repetition of key steps to enhance the learning process. The coverage of solution techniques is
limited to avoid confusion at the initial stages of learning. To better integrate the material, the same
examples are used to introduce diﬀerent techniques. Altogether, just three examples are employed,
one on the optimal extraction of a natural resource, one on consumer utility maximization, and the
final example covers a simple real business cycle model, which can be considered an extension of
the utility maximization example.
Every example is accompanied by Maple computer code to make it possible for readers not
only to replicate the given example but also to extend it in various directions and to encourage
application of the techniques to other areas in economics. As the next step in the process of
learning dynamic programming techniques, it is recommended to study and solve the economic
examples discussed in Add and Cooper (2003). Some of them will seem immediately familiar based
on the coverage in this paper. Others will open up new avenues of applying dynamic programming
techniques.

